8.962 Pset 5 Solutions

Here we have a function which computes the Christoffel symbols given a metric.

```
In[1]:= MakeΓ = Function[{coord, g},
          With[{gInv = Inverse[g] }, Table[1 / 2 Sum[gInv[[α, κ]] (D[g[[β, κ]], coord[[γ]]] +
                 D[g[[γ, κ]], coord[[β]]] - D[g[[β, γ]], coord[[κ]]]), {κ, 1, Length[coord]}],
            {α, 1, Length[coord]}, {β, 1, Length[coord]}, {γ, 1, Length[coord]}]]]
```

```
Out[1]= Function[{coord, g}, With[{gInv = Inverse[g]},
          Table[1/2 Sum_{κ=1}^{Length[coord]} gInv⟦α, κ⟧ (∂_coord⟦γ⟧ g⟦β, κ⟧ + ∂_coord⟦β⟧ g⟦γ, κ⟧ - ∂_coord⟦κ⟧ g⟦β, γ⟧),
           {α, 1, Length[coord]}, {β, 1, Length[coord]}, {γ, 1, Length[coord]}]]]
```

And here we compute the curvature tensor given the connection:

```
In[2]:= MakeR = Function[{coord, Γ}, With[{n = Length[coord]},
           Table[D[Γ[[a, d, b]], coord[[c]]] - D[Γ[[a, c, b]], coord[[d]]] +
             Sum[Γ[[a, c, e]] Γ[[e, d, b]], {e, 1, n}] - Sum[Γ[[a, d, e]] Γ[[e, c, b]], {e, 1, n}],
            {a, 1, n}, {b, 1, n}, {c, 1, n}, {d, 1, n}]]]
```

```
Out[2]= Function[{coord, Γ}, With[{n = Length[coord]},
          Table[∂_coord⟦c⟧ Γ⟦a, d, b⟧ - ∂_coord⟦d⟧ Γ⟦a, c, b⟧ + Sum_{e=1}^{n} Γ⟦a, c, e⟧ Γ⟦e, d, b⟧ -
            Sum_{e=1}^{n} Γ⟦a, d, e⟧ Γ⟦e, c, b⟧, {a, 1, n}, {b, 1, n}, {c, 1, n}, {d, 1, n}]]]
```

Problem 1

▪ a

Here are the Christoffel symbols in spherical coordinates, where we have inserted an ϵ to keep track of powers of Φ:

```
In[3]:= Γsph =
         FullSimplify[Series[MakeΓ[{t, r, θ, ϕ}, DiagonalMatrix[{-(1 + 2 ϵ Φ[r]), 1 - 2 ϵ Φ[r],
              r^2 (1 - 2 ϵ Φ[r]), r^2 Sin[θ]^2 (1 - 2 ϵ Φ[r])}]], {ϵ, 0, 1}]]
```

Out[3]= $\{\{\{0, \Phi'[r]\,\epsilon + O[\epsilon]^2, 0, 0\}, \{\Phi'[r]\,\epsilon + O[\epsilon]^2, 0, 0, 0\}, \{0, 0, 0, 0\}, \{0, 0, 0, 0\}\},$
$\{\{\Phi'[r]\,\epsilon + O[\epsilon]^2, 0, 0, 0\}, \{0, -\Phi'[r]\,\epsilon + O[\epsilon]^2, 0, 0\},$
$\{0, 0, -r + r^2\,\Phi'[r]\,\epsilon + O[\epsilon]^2, 0\}, \{0, 0, 0, -r\,\mathrm{Sin}[\theta]^2 + r^2\,\mathrm{Sin}[\theta]^2\,\Phi'[r]\,\epsilon + O[\epsilon]^2\}\},$
$\{\{0, 0, 0, 0\}, \{0, 0, \frac{1}{r} - \Phi'[r]\,\epsilon + O[\epsilon]^2, 0\}, \{0, \frac{1}{r} - \Phi'[r]\,\epsilon + O[\epsilon]^2, 0, 0\},$
$\{0, 0, 0, -\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\}\}, \{\{0, 0, 0, 0\}, \{0, 0, 0, \frac{1}{r} - \Phi'[r]\,\epsilon + O[\epsilon]^2\},$
$\{0, 0, 0, \mathrm{Cot}[\theta]\}, \{0, \frac{1}{r} - \Phi'[r]\,\epsilon + O[\epsilon]^2, \mathrm{Cot}[\theta], 0\}\}\}$

Here we show that $-(\Gamma^\theta)_{\mu\nu}\, u^\mu\, u^\nu = 0$ for equatorial orbits which have $u^\theta = 0$ initially:

```
In[4]:= FullSimplify[
         With[{u = {ut, ur, 0, uϕ}}, Sum[-Γsph[[3, m, n]] u[[m]] u[[n]], {m, 1 , 4}, {n, 1, 4}]]]
```

Out[4]= $u\phi^2\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]$

```
In[5]:= % /. θ → Pi / 2
```

Out[5]= 0

■ b

Here we compute the RHS of the r-component of the geodesic equation in an equatorial orbit, $-(\Gamma^r)_{\mu\nu}\, u^\mu\, u^\nu$

```
In[6]:= FullSimplify[With[{u = {ut, 0, 0, uϕ}},
           Sum[-Γsph[[2, m, n]] u[[m]] u[[n]], {m, 1, 4}, {n, 1, 4}]] /. θ → Pi / 2]
```

Out[6]= $r\,u\phi^2 - (ut^2 + r^2\,u\phi^2)\,\Phi'[r]\,\epsilon + O[\epsilon]^2$

Specializing to the particular form of the potential we have, we can solve for uϕ in terms of ut (we need to remove the power-series with Normal[...] or else *Mathematica* will try to satisfy this order-by-order in ϵ, which we do not want):

```
In[7]:= Solve[FullSimplify[Normal[%] /. Φ → Function[r, -G M / r]] == 0, uϕ] // FullSimplify
```

Out[7]= $\left\{\left\{u\phi \to -\frac{\sqrt{G}\,\sqrt{M}\,ut\,\sqrt{\epsilon}}{\sqrt{r^2\,(r - G\,M\,\epsilon)}}\right\}, \left\{u\phi \to \frac{\sqrt{G}\,\sqrt{M}\,ut\,\sqrt{\epsilon}}{\sqrt{r^2\,(r - G\,M\,\epsilon)}}\right\}\right\}$

Here is a power series in ϵ for $\Omega \equiv u\phi/ut$

```
In[8]:= FullSimplify[Series[(uϕ /. %[[2]]) / ut, {ϵ, 0, 2}], {r > 0, G > 0, M > 0}]
```

Out[8]= $\sqrt{\frac{G\,M}{r^3}}\,\sqrt{\epsilon} + \frac{(G\,M\,r)^{3/2}\,\epsilon^{3/2}}{2\,r^4} + O[\epsilon]^{5/2}$

We see that to leading order in ϵ, $\Omega = \sqrt{\frac{G M}{r^3}}$, which is the Keplerian orbital frequency. Note that, technically speaking, it is wrong to even include the higher order terms, since our line element is only accurate to leading order in ϵ.

■ c

We need the connection coefficients and curvature in Cartesian coordinates

```
In[9]:= Γ = FullSimplify[Series[MakeΓ[{t, x, y, z}, DiagonalMatrix[{-1, 1, 1, 1}] -
            DiagonalMatrix[2 ϵ Φ[Sqrt[x^2 + y^2 + z^2]] {1, 1, 1, 1}]], {ϵ, 0, 1}]];
```

```
In[10]:= R = FullSimplify[Series[Normal[MakeR[{t, x, y, z}, Γ]], {ϵ, 0, 1}]];
```

(The output of this is really long, so we won't display it here.)

Now we will exploit that the trajectory of the orbit is x = RR Cos[Ω t], y = RR Sin[Ω t], z = 0, with $\tau \sim$ t to lowest order in the velocity and compute the geodesic deviation equation:

```
In[11]:= FullSimplify[Series[Normal[With[{trajSubs = {x -> RR Cos[Ω t], y -> RR Sin[Ω t], z → 0}},
          With[{Γ = Γ /. trajSubs, R = R /. trajSubs, ξ = {ξt[t], ξx[t], ξy[t], ξz[t]}},
           With[{Γm = Table[Γ[[m, 1, n]], {m, 1, 4}, {n, 1, 4}],
             Rm = Table[R[[m, 1, 1, n]], {m, 1, 4}, {n, 1, 4}]},
            D[ξ, {t, 2}] + 2 Γm.D[ξ, t] - Rm.ξ]]]], {ϵ, 0, 1}], RR > 0]
```

Out[11]= $\Big\{\xi t''[t] + 2\,(\mathrm{Cos}[t\,\Omega]\,\xi x'[t] + \mathrm{Sin}[t\,\Omega]\,\xi y'[t])\,\Phi'[RR]\,\epsilon + O[\epsilon]^2,$
$\xi x''[t] + \frac{1}{2\,RR}\big(((2\,\mathrm{Sin}[t\,\Omega]^2\,\xi x[t] - \mathrm{Sin}[2\,t\,\Omega]\,\xi y[t] + 4\,RR\,\mathrm{Cos}[t\,\Omega]\,\xi t'[t])\,\Phi'[RR] +$
$RR\,(2\,\mathrm{Cos}[t\,\Omega]^2\,\xi x[t] + \mathrm{Sin}[2\,t\,\Omega]\,\xi y[t])\,\Phi''[RR])\,\epsilon\big) + O[\epsilon]^2,$
$\xi y''[t] + \frac{1}{2\,RR}\big(((-\mathrm{Sin}[2\,t\,\Omega]\,\xi x[t] + 2\,\mathrm{Cos}[t\,\Omega]^2\,\xi y[t] + 4\,RR\,\mathrm{Sin}[t\,\Omega]\,\xi t'[t])\,\Phi'[RR] +$
$RR\,(\mathrm{Sin}[2\,t\,\Omega]\,\xi x[t] + 2\,\mathrm{Sin}[t\,\Omega]^2\,\xi y[t])\,\Phi''[RR])\,\epsilon\big) +$
$O[\epsilon]^2,\ \xi z''[t] + \frac{\xi z[t]\,\Phi'[RR]\,\epsilon}{RR} + O[\epsilon]^2\Big\}$

Two things to note about the above (which is a pretty complicated expression): 1) We evaluate all tensors (Γ and R) on the trajectory, so everything is a function of t when we start taking derivatives. 2) Γm and Rm are the matrices $(\Gamma^m)_{0\,n}$ and $(R^m)_{0\,0\,n}$ which act on ξ^n.

Problem 2

We need the connection coefficients to first order in Φ:

```
In[12]:= Γ = FullSimplify[Series[MakeΓ[{t, x, y, z}, DiagonalMatrix[{-1, 1, 1, 1}] -
            2 ϵ Φ[Sqrt[x^2 + y^2 + z^2]] DiagonalMatrix[{1, 1, 1, 1}]], {ϵ, 0, 1}]];
```

The output here is the same as from the above problem, so we won't display it again.

Here is the sum on the RHS of the first-order geodesic equation for u^y (note that we evaluate this on the trajectory z = 0, y = b, t = x):

```
In[13]:= FullSimplify[With[{u0 = {1, 1, 0, 0}},
           -Sum[Γ[[3, m, n]] u0[[m]] u0[[n]], {n, 1, 4}, {m, 1, 4}]] /. {y → b, z → 0, t → x}]
```

Out[13]= $-\frac{2\,\left(b\,\Phi'\left[\sqrt{b^2+x^2}\,\right]\right)\,\epsilon}{\sqrt{b^2+x^2}} + O[\epsilon]^2$

```
In[14]:= Integrate[Normal[%] /. Φ → Function[r, -G M / r],
          {x, -Infinity, Infinity}, Assumptions → {b > 0}]
```

Out[14]= $-\frac{4 G M \epsilon}{b}$

Problem 3

Here are the connection coefficients for the sphere of radius a.

```
In[15]:= MatrixForm[Γ = FullSimplify[MakeΓ[{θ, ϕ}, a^2 DiagonalMatrix[{1, Sin[θ]^2}]]]]
```

Out[15]//MatrixForm=

$$\begin{pmatrix} \begin{pmatrix} 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ -\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta] \end{pmatrix} \\ \begin{pmatrix} 0 \\ \mathrm{Cot}[\theta] \end{pmatrix} & \begin{pmatrix} \mathrm{Cot}[\theta] \\ 0 \end{pmatrix} \end{pmatrix}$$

Problem 4

■ a

Here we compute $R_{i\,j\,k\,l}$ (which requires lowering an index):

```
In[16]:= coord = {θ, ϕ}
```

Out[16]= {θ, ϕ}

```
In[17]:= g = a^2 DiagonalMatrix[{1, Sin[θ]^2}]
```

Out[17]= $\{\{a^2, 0\}, \{0, a^2\,\mathrm{Sin}[\theta]^2\}\}$

Here it is:

```
In[18]:= MatrixForm[FullSimplify[With[{R = MakeR[coord, MakeΓ[coord, g]]},
           Table[Sum[g[[i, ip]] R[[ip, j, k, l]], {ip, 1, 2}],
            {i, 1, 2}, {j, 1, 2}, {k, 1, 2}, {l, 1, 2}]]]]
```

Out[18]//MatrixForm=

$$\begin{pmatrix} \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix} & \begin{pmatrix} 0 & a^2\,\mathrm{Sin}[\theta]^2 \\ -a^2\,\mathrm{Sin}[\theta]^2 & 0 \end{pmatrix} \\ \begin{pmatrix} 0 & -a^2\,\mathrm{Sin}[\theta]^2 \\ a^2\,\mathrm{Sin}[\theta]^2 & 0 \end{pmatrix} & \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix} \end{pmatrix}$$

As we expect, there is only one independent component for R.

▪ b

For u=(dθ,0) and v=(0,dϕ), we have $(R^i)_{jkl}\, u^k\, u^l$ (note that I've introduced ϵs in order to keep track of powers of the area):

```
In[19]:= MatrixForm[T = {{1, 0}, {0, 1}} +
            With[{u = {Sqrt[ϵ] dθ, 0}, v = {0, Sqrt[ϵ] dϕ}, R = MakeR[coord, MakeΓ[coord, g]]},
             FullSimplify[Table[Sum[R[[i, j, k, l]] u[[k]] v[[l]],
                {k, 1, 2}, {l, 1, 2}], {i, 1, 2}, {j, 1, 2}]]]]
```

```
Out[19]//MatrixForm=
```

$$\begin{pmatrix} 1 & \mathrm{d}\theta\, \mathrm{d}\phi\, \epsilon\, \mathrm{Sin}[\theta]^2 \\ -\mathrm{d}\theta\, \mathrm{d}\phi\, \epsilon & 1 \end{pmatrix}$$

Now we check that, to lowest order this transformation preserves the metric, and therefore lengths:

```
In[20]:= FullSimplify[Normal[Series[Transpose[T].g.T, {ϵ , 0, 1}]] - g]

Out[20]= {{0, 0}, {0, 0}}
```

Here is $\frac{A\cdot(T\,A)}{|A|\,|T\,A|}$ (note that we carry the series out to order ϵ^2because we expect θ~ϵ, and the angle difference ~ Cos[θ] = 1 - θ^2 / 2).

```
In[21]:= FullSimplify[With[{A = {Aθ, Aϕ}},
           Series[(A.g.(T.A)) / (Sqrt[(A.g.A)] Sqrt[(T.A).g.(T.A)]), {ϵ, 0, 2}]]]
```

$$\text{Out[21]= } 1 - \frac{1}{2}\left(\mathrm{d}\theta^2\, \mathrm{d}\phi^2\, \mathrm{Sin}[\theta]^2\right) \epsilon^2 + O[\epsilon]^3$$

We see that θ = Sin[θ] dθ dϕ.

Problem 6

▪ a

Here's the metric:

```
In[22]:= g = DiagonalMatrix[{-Exp[2 ϕ[x]], Exp[-2 ψ[x]]}]
```

$$\text{Out[22]= } \{\{-e^{2\,\phi[x]}, 0\}, \{0, e^{-2\,\psi[x]}\}\}$$

```
In[23]:= R = MakeR[{t, x}, MakeΓ[{t, x}, g]] // FullSimplify
```

$$\text{Out[23]= } \{\{\{\{0, 0\}, \{0, 0\}\}, \{\{0, -\phi'[x]^2 - \phi'[x]\,\psi'[x] - \phi''[x]\}, \{\phi'[x]\,(\phi'[x] + \psi'[x]) + \phi''[x], 0\}\}\},$$
$$\{\{\{0, -e^{2\,(\phi[x]+\psi[x])}\,(\phi'[x]\,(\phi'[x] + \psi'[x]) + \phi''[x])\},$$
$$\{e^{2\,(\phi[x]+\psi[x])}\,(\phi'[x]\,(\phi'[x] + \psi'[x]) + \phi''[x]), 0\}\}, \{\{0, 0\}, \{0, 0\}\}\}\}$$

Now if we lower an index, we will find the non-zero component (there's only one in 2-D) of R.

```
In[24]:= Table[Sum[g[[i, ip]] R[[ip, j, k, l]], {ip, 1, 2}],
           {i, 1, 2}, {j, 1, 2} , {k, 1, 2}, {l, 1, 2}] // FullSimplify

Out[24]= {{{{0, 0}, {0, 0}},
           {{0, e^(2 φ[x]) (φ'[x] (φ'[x] + ψ'[x]) + φ''[x])}, {-e^(2 φ[x]) (φ'[x] (φ'[x] + ψ'[x]) + φ''[x]), 0}}},
          {{{0, -e^(2 φ[x]) (φ'[x] (φ'[x] + ψ'[x]) + φ''[x])}, {e^(2 φ[x]) (φ'[x] (φ'[x] + ψ'[x]) + φ''[x]), 0}},
           {{0, 0}, {0, 0}}}}
```